

2차원 불연속체 해석 프로그램(UDEC)

1 해석목적

- 해석/설계 측면에서 볼 때 터널이 굴착되는 지반상태 혹은 종류에 따라 지반을 불연속면의 특성을 주요 대상으로 한 불연속체 모델로 가정하여 접근하는 방법을 도입할 필요성 대두
- 특히 절리, 편리, 면리 등을 포함하고 있는 암반의 경우 암반자체의 물리, 역학적 특성보다는 암반 내에 존재하는 절리의 거동특성에 의해 터널안정성에 더욱 영향을 미침
- N.Barton의 연구결과 Q값이 0.01 ~ 10의 범위에서는 암반을 불연속체 모델로 적용하는 것이, 그 외의 Q값에서는 연속체로 모델화하는 것이 보다 현실적이며 실제 상황에 잘 부합된다고 설명하고 있음
- 적용된 지보패턴의 적정성을 확인하기 위해 취약단면을 대표단면으로 선정하고 지반조건, 하중조건, 시공방법 및 시공순서에 따라 해석을 수행하여 터널 주변지반의 변형, 지보재에 발생하는 응력을 검토
- 해석을 통해 굴착공법의 타당성을 검토하고 해석결과와 연관된 계측계획 수립에 활용

2 해석방법

● 지보패턴구분

구 분	암 반 등 급				
	I	II	III	IV	V
개 념 도					
RMR	100 ~ 81	80 ~ 61	60 ~ 41	40 ~ 21	20 ~ 0
Q	40 이상	40 미만 ~ 4	4 미만 ~ 1	1 미만 ~ 0.1	0.1 미만
해석방법	연속체 해석	연속체 해석, 불연속체해석			연속체해석

- 암반등급 II, III, IV에 대하여 비선형 거동이 모사 가능한 Barton-Bandis 모델에 의한 해석을 실시하여 설계의 적정성 검증

3 개별요소법(DEM)의 해석특징


- DEM은 절리와 같은 불연속물질의 정적거동과 동적거동을 구현할 수 있는 방법이며 지하수, 열전달과도 연계(Coupling)되어 사용될 수 있다. 특히 불연속면의 대변위 거동을 예측 할 수 있으며, 이는 다른 수치해석 기법과 명확히 구분되는 차이라고 볼 수 있다.
- 불연속면들은 각 암반블록 사이의 경계로서 구현되며 블록은 강체거동 또는 변형거동이 가능하다. UDEC은 불연속면과 블록의 대변위 거동을 예측하기 위해 'Lagrangian Method'를 사용하고 있다.
- FEM, FDM, BEM에서도 불연속면을 어느정도 표현할 수 있는 모델을 개발하였지만 일반적으로 다음과 같은 문제점을 포함하고 있다.
- 많은 절리가 분포할 경우 이를 구현하기가 매우 어렵다.
- 절리와 절리에 의해서 형성되는 불연속면의 거동을 분석하기 어렵다.
- 변형과 회전이 미소변형의 경우에 한정된다.

이에 비하여 DEM 해석은 일반적으로 다음과 같은 특징을 가지고 있다.

- 불연속면의 변형, 회전, 블록의 이탈을 구현할 수 있다.
- 절리군에 의해 형성되는 접촉면을 해석할 수 있다.

4 UDEC의 이론적 배경

<그림 1>은 UDEC의 해석기법인 개별요소법에서 사용되는 블록과 절리사이의 계산 과정 모식도를 나타내고 있다. 변형체블록은 여러개의 유한한 영역으로 세분화되어 2차원 연속체로 모사되고, 절리는 블록과 블록사이에 매개하는 스프링-슬라이더의 탄소성 1차원 모델로 모사된다.



<그림 1> 개별요소법을 위한 계산 과정

5 블록의 구성모델

UDEC에서 적용되는 블록의 모델은 <표 1>과 같다.

<표 1> UDEC 블록모델

구성 모델	대표적 재료	적용 예
Null	-	공동, 굴착공간의 모델링
Elastic, Isotropic	균일 등방성 연속체 : 선형 응력/변형을 거동	제조된 재료(강철) 안전틀의 계산
Drucker-Prager	낮은 내부마찰각을 가지는 소성물질	연약지반
Mohr-Coulomb Plasticity	느슨하거나 고결된 입상재료(토사, 암, 콘크리트)	토질 및 암반역학 (사면안정, 지하굴착)
Ubiquitous Joint	강도 이방성을 보이는 층상재료	조밀하게 발달한 층리지반의 굴착
Strain-hardening/softening	재료의 비선형 경화/연화를 보이는 입상재료	소성경화 및 잔류강도의 연구(점진파괴)
Double Yield	압력에 의해 영구 체적 압축을 받는 입상 재료	사력댐의 침하검토

● 불연속체 모델 (Mohr-Coulomb)

구 분		개 요	적 용
불연속체 모델	Mohr-Coulomb 모델	- 전단강도식 $\tau_f = \sigma \tan \phi + c$ - 탄 · 소성모델 - Dilation을 상수로 취급하여 전단에 의한 부피 변화 무시	◎

6 절리의 구성모델

<표 2> 절리구성모델

구성 모델	특 징
Continuously Yielding	• 절리의 비선형거동을 표현할 수 있는 연속 항복모델로 일반적인 절리나 파쇄대를 모델링에 적용
Barton-Bandis	• 전단강도식 $\tau_f = \sigma'_n \tan \left(\phi_b + JRC \log_{10} \left(\frac{JCS}{\sigma_n} \right) \right)$ • joint normal behavior • 응력-변형률 경로 : hyperbolic 형태 • 재하와 제하의 반복시 응력이력 고려 • 연속적인 재하와 제하의 반복시 normal stiffness의 증가 • 전단변형으로 인한 절리면의 이탈시 normal stiffness의 변화 고려 • 절리의 밀착도와 거칠기에 의해 응력 경로의 정점이 결정 • joint shear behavior • 연직응력과 전단변형의 함수로서 dilation이 표현 • 최대전단이력으로 인한 절리의 손상도 표현 • 최대전단이력의 반전시 2차 최대전단응력의 감소 • 불연속면의 변형과 강도에 의한 표면거칠기 효과 고려 가능 • 비선형 수직거동, 비선형 전단거동 고려 가능 • 실내시험의 결과에 대한 치수효과 고려 가능 • 전단에 의한 암반의 부피변화 고려 가능
Joint Shear	• 연직응력과 전단변형의 함수로서 dilation이 표현 • 최대전단이력으로 인한 절리의 손상도 표현 • 최대전단이력의 반전시 2차 최대전단응력의 감소

● 불연속체 해석 설계정수의 산정방법(Barton-Bandis Model 중심으로)

구 분	산 정 방 법
불연속면의 방향성	•지표노두조사 및 BIPS, Televue 시뮬결과 이용하여 주절리 방향 산정
위경사 환산	•대표불연속면군의 분포 확인후 터널 축방향에 대한 보정 실시
불연속면 거칠기, JRC (Joint Roughness Coeff.)	•Barton(1983)이 제시한 정성적인 방법 •실험실 최대전단강도로부터 역산 $JRC = \frac{\tan^{-1}\left(\frac{\tau}{\sigma_n}\right) - \phi_b}{\log_{10}\left(\frac{JCS}{\sigma_n}\right)}$
불연속면 압축강도, JCS (Joint Compressive Strength)	•Miller(1965)가 제시한 현장 Schmidt Hammer Test, 암석의 건조밀도를 이용하여 산정 $JCS \approx 10^{(0.88 \times \gamma \times R + 1.01)}$ 여기서, JCS, γ : 단위 MPa R : Schmidt 해머 (L-hammer) 반발계수
초기 불연속면 간극 (e_i)	•산정된 JRC, JCS로부터 초기절리간극 선정 $e_i = \frac{JRC}{5} \left(0.2 \frac{\sigma_c}{JCS} - 0.1 \right)$
불연속면 수직강성 (K_n) 불연속면 전단강성 (K_s)	•초기불연속면 간극으로부터 초기수직강성(K_{ni}) 및 최대달힘변위(Δv_{max})를 구한 후 Barton, Bandis가 제시한 다음 식으로부터 산정 $K_n = K_{ni} \left[1 - \frac{(\sigma_n - \sigma_{ni})}{(\sigma_n - \sigma_{ni}) + K_{ni} \Delta v_{max}} \right]^{-2}$ $K_{ni} = -7.15 + 1.75JRC + 0.02 \left(\frac{JCS}{e_i} \right)$ $\Delta v_{max} = A + B \times JRC_0 + C \left(\frac{JCS_0}{e_i} \right)^D$ $K_s = \frac{\sigma_n}{0.01L} \tan \left[\phi_r + JRC \log_{10} \left(\frac{JCS}{\sigma_n} \right) \right]$

- UDEC해석시 암반의 항복거동을 표현하기 위해 블록모델로는 Mohr-Coulomb Plasticity Model을 절리 구성모델은 Coulomb Slip Model을 적용하였다.

- 주향(Strike) : 경사진 지층면(사면)이 수평면과 이루는 교선의 방향이며 보통 주향선이 정북방향과 이루는 예각(N30°E, N15°W 등)으로 표현하거나 주향선이 정북방향과 시계방향으로 이루는 각도(070, 145 등)로 표현한다.

※ 편각에 의해서 우리나라는 진북(지도상)과 자북이 6.5W 정도 차이가 난다. 그래서 주향과 경사를 측정한 후에 진북으로 보정을 해줘야 한다.

ex) N40E, 30SE $\Rightarrow$ N33E, 30SE



- 진경사(True Dip) : 주향에 직각으로 켜진 사면의 각으로 표면의 최대 경사각임
- 위경사(Apparent Dip) : 주향선에서 수직 단면이 아닌 임의 단면에서 측정된 경사각으로 항상 진경사 > 위경사 이다.

※ 진경사를 위경사로 바꾸는 방법

자북과 진북(지도상)에 대하여 반시계 방향으로 보정 및 터널 방향에 대하여 보정

$$\tan \text{위경사} = \tan \text{진경사} \times \sin \beta$$

$$\tan \text{위경사} = \tan \text{진경사} \times \cos \beta_1$$



[주향과 경사 측정]

예제)

1. 진경사가 30°이고 주향선과 단면적이 이루는 각이 45°일 때 위경사는?

$$\tan \theta = \tan \alpha \times \sin \beta$$

(단, θ 는 위경사, α 는 진경사, β 는 주향과 단면적이 이루는 각)

$$\tan \theta = \tan 30^\circ \times \sin 45^\circ = 0.408$$

$$\theta = \tan^{-1} 0.408 = 22^\circ$$


☒ 해석의 수행을 위하여 먼저 dos memory 확장이 필요하며 방법은 다음과 같다.

- udec program의 등록정보 상에서 udec.exe 뒤에 메모리량을 적어준다
예)c:\WITASCAWUDECWudec.exe 64
- 다른 폴더에서 작업을 원할 경우 udec.bat을 작업폴더로 옮기고 마우스 우측버튼을 누른 후 편집을 선택하고 98일 경우 다음과 같이 경로를 지정하여 준다. C:\WITASCAWUDECWudec.exe 64 윈도우체계가 2000일 경우는 udec.exe 뒤에 64만 적어 주면 됨
- udec.bat이 읽기 전용으로 되어 있으면 등록정보상에서 읽기전용 check를 지움

[상하반단면 쌍굴터널]

본 예제는 암반등급 4등급에 설치되는 복선터널로 굴진장이1.5m이고, 상·하반 2분할 굴착을 한다. 록볼트는 D25-4m를 1굴착당 15개를 설치한다. 모델링은 아래 그림과 같다.

```
*****
* TYPE - 4
* 암반등급 : 4등급 록볼트=4.0m Shotcrete=0.12m
* 록볼트간격 : 종(C.T.C=1.5m) 횡(C.T.C=1.5m) Ko=0.5 하중분담율 : 50-25-25
*****
```



TITLE

Tunnel Analysis Study

TITLE 명령아래에 주석을 달 경우 한글은 2column 이므로 80column이 넘지 않게 주의 요망 또는 앞에 ; or *표시

set log on

udec.log 파일명으로 log파일을 생성함.
파일명을 바꿀 경우 다음과 같이 적어준다.
예) set log 파일명.log
set log on


SET GRAVITY 0 -9.81

RO 0.035

모서리가 뾰족할 경우 발생하는 블록의 Hanging-up에 의한 응력집중이 발생함 이를 방지하기 위하여 모서리를 둥글게 함으로서 블록의 회전과 대변위를 허용함.

변형 가능한 블록의 d값은 양호한 해를 얻기위해 요소 평균길이(블록 한 변 길이)의 1%에 가깝도록 지켜져야함.

default값은 0.5이며 보통 터널모양의 생성을 위하여 경험상 0.01~0.04정도의 값이 적당하다. ro값이 너무 작으면 터널 모양이 생성되지 않으며 너무 크면 터널 모양이 거칠게 생성됨 또한 해의 부정확성을 초래할수 있음.



BL -78.727,-56.632 -78.727,21.766 -41.190,32.36 -32.040,32.36 &

-26.341,31.36 -15.747,32.36 11.916,32.36 51.444,41.36 78.727,59.038 78.727,-56.632

block : 해석영역의 크기에 맞는 블록의 모서리 좌표를 좌측하단부터 시계방향으로 지정하는 명령으로 기본 메모리에서는 모서리의 개수가 10개 정도로 제안되어지므로 보다 복잡한 지반형상을 묘사하기 위해서는 메모리 확장이 필요함.

BL x1,y1 x2,y2 x3,x3.....,

1 line당 데이터 입력 column수가 80column 제한되어 있으므로 데이터가 제한된 수의 column을 넘길 경우 위와 같이 line의 끝에 & 표시를 하고 line을 넘긴 다음 나머지 데이터를 적어줌.

crack -78.7268,19.7664 -41.1895,30.36

crack -41.1895,30.36 -32.0399,30.36

crack -32.0399,30.36 -26.3407,29.36

crack -26.3407,29.36 -15.7471,30.36

crack -15.7471,30.36 11.9161,30.36

crack 11.9161,30.36 51.4442,39.36

crack 51.4442,39.36 78.7268,57.038

crack 명령을 사용하여 지반층을 생성함.

crack x1,y1 x2,y2

crack 명령은 하나의 직선 crack(불연속면)을 블록 안에 만드는 명령으로 시작점 좌표와 끝점 좌표로 정의 된다.

ARC -15.8111 0.5868 -9.7043 4.1126 120 30

ARC -16.9335 1.2348 -21.918 4.1126 50.873783 12

ARC -14.4037 1.3994 -9.3718 -0.632 51.984568 12

arc명령을 사용하여 좌측 터널형상생성

arc xc yc xb yb theta n

중심점(xc yc), 시작점(xb yb), 회전각(theta), 세그먼트 수(n), 회전각 입력시 +각도는 반시계방향이며 -각도는 시계방향을 의미한다.

*주의

터널 모양생성 시 세그먼트 수를 너무 많이 주면 교착각이 너무 커져 에러발생

CR -22.3113 -0.816 -9.3718 -0.632

crack 명령을 사용하여 좌측터널의 굴착 바닥면을 생성함.

해석 시 굴착 바닥면의 맹암거 굴착부분을 묘사하는 경우가 있는데 유텍에서는 이 부분에 작은 블록이 생성되어 수렴 시 문제가 발생하므로 묘사하지 않는 것이 좋음.

CR -23.0 1.3994 -9.0373 1.3994

좌측터널의 상하반단면 굴착선을 crack명령을 사용하여 생성

상하반단면 굴착선 입력시 UDEC 프로그램상에서 Arc로 생성된 터널 부분과 cad 상의 굴착라인 좌표가 서로 교차하지 않는 경우가 발생하여 터널 형상 및 굴착라인이 생성되지 않는 경우가 발생하므로 상하반단면 굴착라인의 시작점과 끝점을 터널외곽굴착라인 밖으로 조금 더 길게 잡아준다.우측 터널형상 생성

ARC 15.8111 0.5868 21.918 4.1126 120 30

ARC 14.4037 1.3994 9.7043 4.1126 51.984568 12

ARC 16.9335 1.2348 22.3113 -0.816 50.873783 12

우측 터널형상 생성

CR 9.3718 -0.632 22.3113 -0.816

우측터널 굴착 바닥면 생성

CR 9.0373 1.3994 22.8268 1.3994

우측터널 상하반단면 굴착선 생성

;방법 1

crack -56.4257,26.7562 -79.1478,1.0171
crack -64.6558,24.7900 -78.7300,11.8800
crack -33.3458,30.8673 -76.1063,-3.4515
crack -32.2723,31.2248 -78.0744,-20.6109
crack -12.7706,31.4035 -70.0232,-34.3741
crack 23.3308,27.6126 -45.4382,-53.9902
crack 58.9784,34.9035 -18.7727,-54.5511
crack 74.6970,18.3586 17.4363,-49.7839
crack 79.7494,-0.4296 23.6115,-57.9161
crack 74.4163,-22.5830 45.5053,-53.9902
crack 75.8198,-27.3501 48.8736,-56.5140
crack 77.7846,-46.4188 69.9253,-57.3553
crack 79.7494,-51.4664 72.7322,-57.9161
crack 60.1012,-57.0749 80.3108,-48.6622
crack 58.1363,39.6707 -61.9989,-7.4402
crack 0.5949,9.1047 79.1881,36.3056
crack 71.6094,36.3056 10.1384,-32.9586
crack 75.2584,23.1258 -2.7734,-41.3712
crack 79.7494,7.1418 9.2963,-49.7839
crack 71.8901,12.4698 -46.8417,-34.3607
crack 3.9632,29.0147 -47.6837,-30.9956
crack -75.7527,-30.9956 2.5597,0.6920
crack 73.0129,29.8559 -12.8782,-5.4772
crack 17.4363,2.9354 -13.7203,-32.1173
crack -17.6499,-19.4983 -50.2099,-58.1966
crack -17.0885,-12.7682 -72.1037,-59.3182
crack -36.4561,-10.8052 -81.3665,-46.4188
crack -82.4893,-48.9426 -2.4927,-18.0962
crack -9.2292,31.2580 -72.3844,-21.1809
crack 80.3108,-32.9586 19.1204,-57.6357
crack -20.4568,-57.0749 77.2232,-17.5354
crack -77.7175,-15.0116 -31.4037,2.6550
crack 72.7322,-7.1598 29.5060,-57.6357
crack 72.1708,-21.4613 38.4880,-56.2336
crack 69.0832,23.1258 -0.2472,-53.4294
crack -52.7361,-53.4294 68.8025,-5.7577
crack -77.1562,-41.0908 1.9984,-8.2815
crack -2.7734,30.6972 -79.4017,-0.7101
crack -49.3679,16.1153 -81.3665,-12.4878
crack -42.9120,1.5333 -81.3665,-27.9110
crack 81.9478,49.2078 51.3472,37.2114
crack 79.8820,45.8539 43.3420,15.5405
crack 70.5856,52.3036 57.8030,30.8907
crack 17.5111,-54.8772 77.4859,-30.6171
crack 34.9494,-56.0169 76.5081,-35.9901
crack 78.9527,-9.7761 11.8069,-56.9938

;방법 2

;JSET 39.1 0 10 2 0 0 15 30 0 0

;JSET 22 0 10 2 0 5 15 30 0 0

;JSET 48.6 0 10 2 0 0 15 30 0 0

불연속면(joint) 생성방법에는 두가지 방법이 있다.

방법1은 crack명령을 이용하여 일일이 불연속면을 입력

방법2는 jset 명령을 사용하여 자동생성하는 방법으로 다음과 같다.

jset a_m,a_d t_m,t_d g_m,g_d s_m,s_d <x0,y0><ad0><range>

a_m : 절리각(양의 x축을 기준으로 하여 +는 반시계, -는 시계방향으로 입력되어짐)

a_d : 평균에서의 최대편차, 최대편차 내에서 절리 자동생성

t_m : 절리의 단위길이(절리방향)

t_d : 절리 단의길이의 최대편차

g_m : 단위 절리 당 간격(절리방향)

g_d : 단위 절리 당 간격의 최대편차

s_m : 절리 수직방향으로의 간격

s_d : 절리 수직방향 간격의 최대편차

<x0, y0> global 축상에서의 절리 생성 시작점

<ad0> global 축상에서의 joint생성 시 local 편차

<range> joint 생성범위 지정

<x0, y0>, <ad0>, <range>생략가능

joint가 들어간 영역은 자동으로 jreg id 1으로 자동 인식되며 joint영역설정 및 물성치영역 설정 시 사용하면 편리함

예) gen edge 2 jreg 1

JDEL ; JDELETE명령은 교차지점을 초과하여 생성되어진 절리를 지우라는 명령으로 생략가능.

GEN EDGE 1.5

GEN EDGE u range x1,x2 y1,y2

각 블록들을 삼각형의 유한차분 요소로 자동으로 분할하는 명령으로 u 는 삼각형 요소의 최대 모서리 길이임. RANGE 명령을 생략하면 전 BLOCK을 삼각형의 유한 차분 요소로 분할함.

GEN QUAD u 명령은 자동으로 각 블록들을 사각형의 유한차분 요소로 분할하는 명령임.

각 명령에 range를 설정하여 터널부근은 삼각형의 유한차분 요소로 분할하고 그 외 부근은 사각형의 유한차분 요소로 분할한 mesh를 생성하여 해석시간을 단축할 수 있다. 또한 소성영역에 대하여서는 사각형의 유한차분 요소가 더 잘 나타낼 수 있다고 함

pl bl zone 명령으로 확인가능

SET DELC ON

변위에 의한 해석영역의 over를 막아줌
접촉부의 분리정도가 round length 보다 0.55배 이상 클 때 자동으로 제거

DAMP auto

DAMP auto 사용 시 해석수행과정에서 지반 물성 입력데이터에 의해 damping값이 자동계산되어지면
서 해석이 수행되어짐 주로 static or steady-state 문제일 경우 사용
DAMP local 사용 시 입력되어있는 default damping값을 사용하여 해석수행이 auto에 비해 빠름 주로
동적해석 시 사용

DELETE RANGE AREA 5.0E-2

매우 작은 block에서의 응력집중을 막기 위하여 작은 block을 지워준다.
print max : mesh생성 후 가장 작은 block의 크기를 확인.

set back 0

화면상의 배경색을 바꿔줌 0: black

SAVE MESH.SAV

생성된 mesh를 파일명.sav 로 저장
다른곳에 저장하고 싶을때는 다음과 같이 경로 지정
d:\폴더명(경로지정)\파일명.sav

RES MESH.SAV ; restore명령은 mesh.save file을 불러오라는 명령임

restore명령은 mesh.save file을 불러오라는 명령임

;풍화암 초기 물성치

```
PROP mat=3 de 2.000e-3 b 4.90e2 sh 3.19e2 coh 600 fr 60 te 500
CHANGE REGION -78.7268,19.7664 -78.7268,21.7664 -41.1895,32.36 &
-41.1895,30.36 MAT=3 CONS=3
CHANGE REGION -41.1895,30.36 -41.1895,32.36 -32.0399,32.36 &
-32.0399,30.36 MAT=3 CONS=3
CHANGE REGION -32.0399,30.36 -32.0399,32.36 -26.3407,31.36 &
-26.3407,29.36 MAT=3 CONS=3
CHANGE REGION -26.3407,29.36 -26.3407,31.36 -15.7471,32.36 &
-15.7471,30.36 MAT=3 CONS=3
CHANGE REGION -15.7471,30.36 -15.7471,32.36 11.9161,32.36 &
11.9161,30.36 MAT=3 CONS=3
CHANGE REGION 11.9161,30.36 11.9161,32.36 51.4442,41.36 &
51.4442,39.36 MAT=3 CONS=3
CHANGE REGION 51.4442,39.36 51.4442,41.36 78.7268,59.038 &
78.7268,57.038 MAT=3 CONS=3
CHANGE BLOCK 498 MAT=3 CONS=3
```

;연암 초기 물성치

```
PROP mat=4 de 2.300e-3 b 1.961e3 sh 1.275e3 coh 600 fr 60 te 500
CHANGE REGION -78.7268,-56.632 -78.7268,19.7664 -41.1895,30.36 &
-41.1895,-56.632 MAT=4 CONS=3
CHANGE REGION -41.1895,-56.632 -41.1895,30.36 -32.0399,30.36 &
-32.0399,-56.632 MAT=4 CONS=3
CHANGE REGION -32.0399,-56.632 -32.0399,30.36 -26.3407,29.36 &
-26.3407,-56.632 MAT=4 CONS=3
CHANGE REGION -26.3407,-56.632 -26.3407,29.36 -15.7471,30.36 &
-15.7471,-56.632 MAT=4 CONS=3
CHANGE REGION -15.7471,-56.632 -15.7471,30.36 11.9161,30.36 &
11.9161,-56.632 MAT=4 CONS=3
CHANGE REGION 11.9161,-56.632 11.9161,30.36 51.4442,39.36 &
51.4442,-56.632 MAT=4 CONS=3
CHANGE REGION 51.4442,-56.632 51.4442,39.36 78.7268,57.038 &
78.7268,-56.632 MAT=4 CONS=3
```

초기물성치는 점착력과 마찰각 인장력을 크게 준 것으로 초기 수렴과정에서의 과대변형으로 인한 block의 파괴를 막아줌. 초기응력계산에 사용되어지는 나머지 물성치는 정확히 주어야함.

실수 : 과물성치 입력시 마찰각은 제외하는 것이 좋을 듯 접착력과 텐션값 증가만으로도 충분

property mat=n den s b fri(암반의 내부마찰각) c ten

n=물성번호

density, shear modulus(g), bulk modulus(k), friction, cohesion, tension

주의

절리와 블록의 물성번호는 같아도 되나 지보재의 물성번호는 절리 물성번호 및 블록 물성 번호와 같아서는 안 됨 또한 물성번호 1은 기본적인 블록과 절리의 물성번호이므로 pl bl mat로 물성 확인 시 확인이 안되므로 사용하지 않는 것이 좋음

확인명령 pl bl mat 또는 pl bl mat den 등

change region x1,y1 x2,y2 x3,y3 x4,y4 mat=n cons=n

정의 되어진 물성치를 입력하는 명령으로 Udec에서는 기본적으로 사용하는 물질모델은 다음과 같다.

cons=0 ; null model

cons=1 ; elastic model

cons=3 ; mohr-coulomb model

임의의 영역을 지정하는 방법에는 여러 가지 방법이 있으며 기본적인 방법은 다음과 같다.

block n ; 블록번호로 블록 지정

range x1 x2 y1 y2 ; 좌측하단좌표와 우측상단좌표를 이용하여 무게 중심을 포함하는 모든 block선택 직 사각형 및 정사각형 형태로만 선택가능

region x1 y1 x2 y2 x3 y3 x4 y4 ; 사각형 영역으로 무게 중심을 포함하는 모든 block을 선택하는 명령으로 임의의 사각형 모양으로 block을 지정할 수 있음

annulus x y r1 r2 : 반경 r1과 r2 사이에 존재하는 모든 블록 설정 x,y : 중심점 좌표 r1: 내부 반경 r2: 내부반경임

다음과 같은 fish문을 이용하여 물성치를 줄수 있다.

def derive

s_mod = y_mod / (2.0 * (1.0 + p_ratio)) ; g 또는 sh

b_mod = y_mod / (3.0 * (1.0 - 2.0 * p_ratio)) ; K 또는 b

end

예)

set y_mod = 5e8 p_ratio = 0.25

derive

PROP mat=4 de 2.300e-3 b b_mod sh s_mod coh 600 fr 60 te 500

*암반의 단위 밀도(ρ) : 2300 kg/m³임 무차원단위계에서 단위변환에 의한값으로 $\times 100$ 이되어 있으므로 혼돈하지 말기

주의

현 해석에 사용되어진 물성단위는 UNIT 10⁶KG/M³, MN, MPa로 BB모델 사용시는 MPa 단위만 사용 가능 다른 단위를 사용할 경우 초기수렴 시 엄청난 변위와 함께 계산 에러 발생.

초기 과물성치 사용시에는 다른단위사용 가능?

물성치 입력시 물성치 계산 에러가 발생하는 경우가 있음?

물성치가 세는 에러 발생시 다시 한번 물성치를 줌?

참고

다음과 같은 순서로 block 번호를 이용하여 물성을 부여하면 보다 손쉽게 물성을 부여할 수 있다.

;지반분류선 생성

```
crack -78.7268,19.7664 -41.1895,30.36
crack -41.1895,30.36 -32.0399,30.36
crack -32.0399,30.36 -26.3407,29.36
crack -26.3407,29.36 -15.7471,30.36
crack -15.7471,30.36 11.9161,30.36
crack 11.9161,30.36 51.4442,39.36
crack 51.4442,39.36 78.7268,57.038
```

;위와 같이 crack 명령으로 두개의 블록 생성후 **pl bl num** 명령으로 두개 block번호 확인 후 블록 번호를 이용하여 아래와 같이 물성치 입력

;풍화암 초기 물성치 입력

```
PROP mat=3 de 2.000e-3 b 4.90e2 sh 3.19e2 coh 600 fr 60 te 500
CHANGE bl=343 MAT=2 CONS=3
```

;연암 초기 물성치 입력

```
PROP mat=4 de 2.300e-3 b 1.961e3 sh 1.275e3 coh 600 fr 60 te 500
CHANGE bl=2 MAT=3 CONS=3
```

;터널형상생성

```
.....
.....
.....
.....
```

;-----;

;초기화

;-----;

```
RESET HIS
RESET VEL
RESET DISP
RESET JDIS
RESET ROT
```

;-----;

;원지반물성치

;-----;

;풍화암

```
PROP mat=3 de 2.000e-3 b 4.90e2 sh 3.19e2 coh 4.900e-2 fr 30 te 2.45e-2
```

;연암

```
PROP mat=4 de 2.300e-3 b 1.961e3 sh 1.275e3 coh 0.324 fr 33 te 0.162
```

본 예제에서는 이와 같은 순서로 물성치를 입력하는 것이 잘 안되어서 사용하지 않았으므로 확인 요망?

CHANGE JMAT=1 JCON=7

change jmat=n1 jcon=n2 range angle a1 a2

해석영역 안에 있는 모든 절리에 JOINT MATERIAL 물성을 부여 함.

range angle a1 a2 ; $a1 \leq x \leq a2$ 범위 안에 존재하는 모든 조인트 선택

range 생략 시 모든 joint 영역에 물성 부여

n1=물성번호 입력

n2는 물질모델명으로 Joint에 대하여 UDEC에서 지원하는 절리모델은 다음과 같다.

n1=1: Point Contact Coulomb Slip Model

n1=2: Coulomb slip model

n1=3: Continuous Yielding Joint Model

n1=5: Continuously yielding

n1=7: Barton-Bandis joint model

또는 모델명으로 입력 가능(예: jcon=bb)

prop jmat=1 jkn=3.0e6 jks=2.0e5 jrc=20 jcs=200.0 phi=60 &
si=700 ln=1 lo=0.1

fictitious joint의 물성을 정의

터널 형상이나 굴착선, zoning 밀도를 다르게 주기 위한 선 등을 fictitious joint라하며, 이들 절리는 암반 블록과 같이 변형해야 하기 때문에 glue joint로 만들어야 함. 보통 물성치보다 수직강성 및 전단강성을 크게 줌 인공적으로 생성시킨 터널형상 절리면에서의 파괴를 막아줌 그러나 fictitious joint의 물성치를 너무크게주면 초기수렴시 수렴시간이 많이 걸리므로 다음식에 의하여 주어짐.

$$j_{kn}=j_{ks}=10 \times \frac{K + \frac{4}{3}G}{\Delta Z_{\min}}$$

K,G는 절리와 인접한 암반블록의 체적 및 전단탄성계수이고 $\Delta Z_{\min}$ 은 Zone의 최소높이로 한번의 edge길이임.

property jmat=n jkn jks jfric jcoh jten

jkn : 불연속면의 수직강성

jks : 불연속면의 전단강성

jrc : 실험실 시편에 대한 불연속면의 거칠기

jcs : 실험실 시편에 대한 일축압축강도

ln : 현장 scale의 불연속면 길이로 입력하지 않을 경우 모델 내에서 자동계산 됨 (lo의 10배 or ln=lo)

lo : 실험실 시험편의 불연속면 길이

phir : 불연속면의 잔류마찰각

sigmac : 무결암에 대한 일축압축강도

plot mat joint 명령으로 확인

*주의 : 간혹 jcs값을 sig값보다 크게 입력하여 error 발생.

*주의 : 마찰각에 대하여 BB모델은 잔류마찰각이며, Mohr Column모델은 기본 마찰각임

CHANGE JMAT=2 JCON=7 RANGE ANGLE 38.1 40.1

해석영역 안에 존재하는 JOINT중에서 38.1도 이상 ~ 40.1도 이하 영역 안에 존재하는 JOINT 에 JOINT MATERIAL 2번의 물성을 부여 함.

change jmat=n1 cons=n2 range angle a1 a2

n1=물성번호

n2=joint 물질모델

range angle a1 a2 ; $a1 \leq x \leq a2$ 범위 안에 존재하는 모든 조인트 선택

range 생략 시 모든 joint영역에 물성 부여

prop jmat=2 jkn=5.16e3 jks=3.54e3 jrc=6 jcs=97.223 phi=26 &
si=300 lo=0.1

joint material 2번의 물성 정의

CHANGE JMAT=3 JCON=7 RANGE ANGLE 21 23

prop jmat=3 jkn=5.96e3 jks=3.54e3 jrc=6 jcs=97.223 phi=26 &
si=196 lo=0.1

CHANGE JMAT=4 JCON=7 RANGE ANGLE 47.6 49.6

prop jmat=4 jkn=5.96e3 jks=3.54e3 jrc=6 jcs=97.223 phi=26 &
si=300 lo=0.1

set jcondf=7

set jcondf = 7(BB) : BB모델로 작업수행 명령

insitu stress -0.362 0 -0.724 ygrad 1.0e-2 0 2.0e-2 szz -0.362 zgrad 0 1.0e-2

초기 수렴과정으로 본 해석에서는 K0=0.5를 사용하였다.

insitu stress는 0,0 지점의 응력값이며 본 해석에서는 0.0을 터널 부근에 설정하였다.

insitu stress sxx sxy sy ygrad n1 n2 n3 szz n1 zgrad szzx sszy range x1 x2 y1 y2

압축응력 부호 (-), 인장응력 부호 (+)

sxx : x방향 초기응력 = $syy \cdot k_0$

sxy : 0

syy : y방향 초기응력 단위중량*높이*9.81

n1 : x방향으로의 응력변화 = 단위중량* k_0 *9.81

n2 : 0

n3 : y방향으로의 응력변화 = 단위중량*9.81

*9.81값은 tf/m^2 단위를 kpa단위로 바꾸어주기 위하여 준 값임. 또한 초기 수렴과정에 있어 초기 응력 값을 너무 크게 주면 수렴이 잘되지 않는 경우가 있으므로 초기 syy값을 구함에 있어 초기 응력값의 1/10 즉 단위중량*높이 값을 적용하고 n3값을 적용함에 있어서는 터널이 존재하는 암반의 단위중량 *9.81값(연암의 단위중량*9.81)보다 한 단계 적은 지반의 물성치(풍화암*9.81)를 사용하는 것이 초기 수렴이 빠름

또는 각각 x,y 방향으로 구역을 나누어 수렴시키는 방법도 있음 여기서 생략

szz: 터널과 같이 plane strain condition인 문제에서는 반드시 szz를 입력하며 탄성해석에서는 필요 없지만 탄소성 해석에서는 파괴에 영향을 주기 때문에 주의 깊게 선택
szzx=0
szzzy: z방향 초기응력 = $syy \cdot k_0$
초기응력설정시

BOUND XVEL=0 RANGE -79.7 -77.7 -57.6 22.8

BOUND YVEL=0 RANGE -79.7 79.7 -57.6 -55.6

BOUND XVEL=0 RANGE 77.7 79.7 -57.6 60

Boundary condition을 부여하는 명령으로 udec에서는 일반적으로 가속도 경계를 사용하며 그 방법은 다음과 같다.

bound xvel=0 range x1 x2 y1 y2

경계면에서의 x방향의 가속도를 0으로 만들어 줌으로서 x방향으로의 변위를 허용하지 않음. 경계조건 롤러와 같으며 모델링의 좌우측경계가 구속되어짐.

좌측경계 시

x1 x2 값은 모델링에서 좌측하단 x좌표값에 $\pm 1m$ 를 한 값이며, y1 값은 좌측하단 y값 좌표이며 y2 값은 좌측상단의 y값 좌표이다.

우측경계 시

x1 x2 값은 모델링에서 우측하단 x좌표값에 $\pm 1m$ 를 한 값이며, y1 값은 우측하단 y값 좌표이며 y2 값은 우측상단의 y값 좌표이다.

bound yvel=0 range x1 x2 y1 y2

경계면에서의 y방향의 가속도를 0으로 만들어 줌으로서 y방향으로의 변위를 허용하지 않음. 모델링의 하부경계구속 조건임.

경계조건이 올바르게 설정되었는지 확인하기 위하여는 pl bound xcond, pl bound ycond 명령으로 확인할 수 있다.

HIST UNBAL ;초기 수렴과정에서 unbalance force를 확인하기 위하여 사용

solve force=1e-3

초기 불균형력의 0.01%이면 수렴한 것으로 보며 solve 명령만 사용하면 자동 설정된 값에 의해 수렴되어짐(force=1e-3생략 가능)

주의

초기수렴 시 상부 지반에 삼각형의 블록이 생성되어져 있을 경우 블록이 경사면을 따라 흐르는 현상 및 공중으로 부양하는 현상에 의하여 수렴이 되지 않는 경우가 있으므로 적당한 cycle(step)을 주어 수렴시키거나 본 예제와 같이 상부 지반에는 crack을 만들지 않는 방법이 있다.

방법 2인 jset 명령을 사용하여 joint를 만들 경우는 range 영역을 연암층 지반상단에서 약간 높게 잡으면 됨 그러나 지표면이 크게 경사져 있을 경우는 range 영역을 잡아주기 힘들.

SAVE INI.SAV ;초기 수렴된 데이터를 ini.sav파일명으로 저장하라는 명령

초기수렴과정에서 블록간의 미끄러짐이 발생하였는지 확인하는 명령으로 초기수렴 시 미끄러움이 발생
되어선 안됨

plot hold bou slip red 명령으로 확인가능

RESET HIS

RESET VEL

RESET DISP

RESET JDIS

RESET ROT

초기화 과정으로 수렴시 발생한 변형을 없애줌으로서 수렴과정에서 생긴 변형이 해석에 영향을 주지
않도록 함

reset hist : 초기 수렴 시 기록 초기화

reset vel : 초기 수렴 시 모든 블록의 가속도 초기화

reset disp : 초기 수렴 시 모든 블록의 변위 초기화

reset jdis : 초기 수렴 시 절리면 변위 초기화

reset rot : 초기 수렴 시 블록의 회전 변위초기화

;풍화암 원지반물성치

PROP mat=3 de 2.000e-3 b 4.900e2 sh 3.190e2 coh 4.900e-2 fr 30 te 2.45e-2

풍화암의 초기 물성치를 원지반 물성치로 바꿈

초기 가물성치 물성치만 원상태로 적어줘도 되는 것으로 알고 있으나 확인 바람

;연암 원지반물성치

PROP mat=4 de 2.300e-3 b 1.961e3 sh 1.275e3 coh 0.324 fr 33 te 0.162

연암의 초기 물성치를 원지반 물성치로 바꿈

SET DELC OFF

*

* 좌측터널 해석 수행

*

call find.fis

call relax.fis

call remove.fis

하중분담 fish를 불러오는 명령으로 아직 실행되어지는 것은 아니며 본 fish가 memory에 기억되어지는 과정임. 본 fish를 작동시키는 방법은 본 fish 내부에 정의되어진 함수명과 입력 변수값을 data에 적어주면 됨.

; <<step 1>>

DELETE REGION -22.68 1.39 -20.07 6.20 -15.81 7.51 -15.81 1.39

DELETE REGION -15.81 1.39 -15.81 7.51 -11.54 6.20 -8.93 1.39

상부반단면 굴착하는 명령으로 상부반단면에 포함되어지는 모든 블록을 지정하여야함

BOUND INTERIOR XVEL=0 YVEL=0 RANGE DOMAIN 1260936

굴착면의 내부경계조건을 구속시켜 주는 명령임

굴착되어진 null모델 및 절리면 사이의 공간을 DOMAIN이라하며 고유번호를 가진다. DOMAIN의 고유번호는 pl bl dn명령으로 확인할 수 있으며 RANGE 명령을 사용하여 Domain영역을 잡아줌으로서 보다 편리하게 굴착되어진 굴착면을 지정할 수 있다.

그러나 Domain number는 mesh 수정에 있어 계속변하는 값이므로 mesh 수정 시 다시 지정해야하는 불편함이 있어 range x1 x2 y1 y2 및 annulus x y r1 r2 영역으로 잡는 경우도 있음 그러나 본 명령어 사용 시 너무 크게 영역을 잡으면 해석결과에 error를 초과하므로 굴착영역에 맞춰 적당히 설정하여야한다.

CYCLE 1 ; 1회의 cycle를 실행시켜 Bound interior.....,명령을 실행시킴

find

하중분담율 fish인 find.fis에서 함수명 find를 실행시킴

본 예제에서의 하중분담율을 50-25-25임 이것은 굴착에 의해 발생한 초기힘을 굴착단계에 따라 50-75-100% 가하는 것으로 본 fish는 초기힘에 역의 힘을 가하는 것으로 입력변수값은 0.5-0.25-0 이된다. find.fis는 다음과 같으며 하중분담율 50%임.

```
def find
  array ini_force(2,1000)
  loop m (1,2)
    loop n (1,1000)
      ini_force(m,n) = 0.0
    endloop
  endloop
  n=1
  ib=block_head ; start of block list
  loop while ib #0 ; loop through all blocks
    igp=b_gp(ib) ; start of gridpoint list for block ib
    loop while igp # 0 ; loop through all gridpoints
      ibou=gp_bou(igp) ; index of boundary corner associated with gridpoint
      if(ibou) < 0 then ; if address is negative then it is interior
        ibou2=abs(ibou)
        if (imem(ibou2+2)) = 4 then ;
          forcex=fmem(ibou2+4) ; total x-force (see page M-10)
          forcey=fmem(ibou2+5) ; total y-force (see page M-10)
          imem(ibou2+2)= 1 ; 터널벽면의 BC를 변위BC에서 하중BC로 바꾸어줌
          imem(ibou2+3)= 1 ; 터널벽면의 BC를 변위BC에서 하중BC로 바꾸어줌
          fmem(ibou2+4)= forcex*0.5 ; apply 50% reaction force
          fmem(ibou2+5)= forcey*0.5 ; apply 50% reaction force
          ini_force(1,n)=forcex
          ini_force(2,n)=forcey
          xforce0=0.0
          yforce0=0.0
          xforce1=0.0
          yforce1=0.0
          xforce0=forcex
          yforce0=forcey
          xforce1=fmem(ibou2+4)
          yforce1=fmem(ibou2+5)
          xtable(xforce0,n)=forcex
          xtable(yforce0,n)=forcey
          xtable(xforce1,n)=fmem(ibou2+4)
          xtable(yforce1,n)=fmem(ibou2+5)
          ii=out( string(n)+'   ibou      '+string(ibou))
          ii=out('n      '+string(n))
          ii=out('xforce0 '+string(xforce0))
          ii=out('yforce0 '+string(yforce0))
          ii=out('ini_fx  '+string(ini_fx(1,n)))
          ii=out('ini_fy  '+string(ini_fy(2,n)))
          ii=out('xforce1 '+string(xforce1))
          ii=out('yforce1 '+string(yforce1))
          n=n+1
        endif
      endif
      igp=gp_next(igp) ; next gridpoint
    endloop
    ib= b_next(ib) ; next block in list
  endloop
end
```

;좌측터널변위

;HIST N=10 ; 10cycle 마다 데이터 저장

HIST YDIS -15.8111 7.6484 ;천단변위

HIST XDIS -22.6967 1.4994 ;왼쪽측벽상반

HIST XDIS -22.6967 1.2994 ; 왼쪽측벽하반

HIST XDIS -8.9673 1.4994 ;오른쪽측벽상반

HIST XDIS -8.9673 1.2994 ;오른쪽측벽하반

HIST YDIS -15.8111 -0.8236 ; 바닥면

;우측터널변위

HIST YDIS 15.8111 7.6484 ; 천단변위

HIST XDIS 8.9673 1.4994 ; 왼쪽측벽상반

HIST XDIS 8.9673 1.2994 ; 왼쪽측벽하반

HIST XDIS 22.6967 1.4994 ; 오른쪽측벽상반

HIST XDIS 22.6967 1.2994 ; 오른쪽측벽하반

HIST YDIS 15.8111 -0.8236 ; 바닥면

history를 지정하는 명령으로 변위를 관측하고 싶은 지점의 좌표를 입력함

his ydis x1 y1 : 지정된 지점의 y변위값을 기록한다.

his xdis x1 y1 : 지정된 지점의 x변위값을 기록한다.

CYCLE 3000

해석을 수행하는 명령으로 보통 3000~5000 cycle이 적당하며 너무 많은 cycle을 줄 경우 절리면에서의 과대 변형이 일어나 해석에 Error 발생. 특히 상부에 킨블록이 형성되어져 있는 경우 킨블록의 미끄럼이 계속발생하여 수렴되지 않음.

수렴시 cycle수가 너무 많으면 해석수행과정에서 overtol문제가 발생 overtol 문제는 한 블록의 모서리가 다른 접촉면 블록의 접촉면으로 겹쳐지는 현상으로 겹쳐진 양이 허용치(d값의 1/10)를 초과하면 발생.

overtol이 발생한 지점을 확인하는 명령은 pl bl overlap임.

보통 overlap 문제의 발생을 막아주는 명령으로 다음과 같은 명령을 data에 추가하는 경우가 있으나 잘못된 결과를 얻을 수 있으므로 사용하지 않는 것이 좋다.

set cscan n ; n 사이클 마다 모서리의 중심점의 좌표를 update하라는 명령 default값은 100임

set overtol n : 겹쳐지는 부분의 허용치를 지정하는 명령으로 default값은 d값의 1/2임

SAVE 1.SAV

; <<step 2>>

STRUCT GEN XC -15.8111 YC 1.4994 NPOINT 100 MAT 7 THI 0.12 FANG 0.01 &
THETA 179.89

struct gen xc yc npoint n thi n mat n fang a theta a connect
xc yc : 생성될 strut의 중심점
npoint n : 생성되는 세그먼트 수로 너무 많이 줄 경우 error발생
thi n : 세그먼트의 두께로 쏫크리트 두께임
mat n : 세그먼트의 물성치 번호로 연한 쏫크리트 물성치를 정의한 물성번호를 적어줌
fang a : 양의 x축을 중심으로 세그먼트가 시작되는 지점까지의 회전각 반시계방향에 +임
theta a : 세그먼트 시작점에서의 세그먼트 끝점까지의 회전각
connect 명령은 쏫크리트(structure)를 연결시켜주는 명령으로 기존 structure의 끝점과 새로 생기는 structure의 시작점을 연결시켜줌 따라서 좌측부와 연결은 이루어지나 우측부는 연결이 안됨 따라서 structure를 약간 겹치는 방식으로 하였음. 확인명령은 **pl bl stru thick red** 임
세그먼트가 원하는 부분에 잘 생성되지 않는 경우는 세그먼트의 중심점을 상하로 조금 이동하거나 fang 및 theta값을 조금 조절하여 준다.
문제점) 쏫크리트가 연결되지 않음으로서 축력이 잘 전달되지 않는 것으로 보임 축력이 측벽부에서 급격히 감소하는 현상발생

prop mat 7 st_dens 2.4e-3 st_prat 0.15 st_ymod 0.5e4
prop mat 7 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6

쏘크리트 물성치
property n st_dens st_prat st_ymod st_ycomp st_yield st_yresid
n 물성번호
st_dens : 단위중량, st_prat : 포아송비 st_ymod : 탄성계수
st_ycomp : 압축항복강도 st_yield : 인장 항복강도 st_yresid : 잔류 인장 항복강도
쏘크리트의 단위중량 : 2.35e-03 ~ **2.5e-03**
포아송 비 : 0.15 ~ **0.2**
탄성계수값은 연한 쏫크리트일 경우 강한 쏫크리트 탄성계수값의 1/3값을 사용
강한쏘크리트 탄성계수 : **1.5e4**

상대적으로 낮은 물성치에 의하여 해석수행과정에서 structure의 파괴를 막기 위하여 본 예제에서는 압축항복강도 인장항복강도 잔류 인장 항복강도를 크게 주었다 이는 본 해석 수행의 목적이 structure의 안정성에 대하여 허용응력으로 검토함에 있기 때문임.

Udec 매뉴얼에서 제공되는 물성치는 다음과 같으며, 생략가능하는 경우도 있음.
연한쏘크리트
prop mat 7 st_ycomp 4 st_yield 2 st_yresid 1
강한쏘크리트
prop mat 7 st_ycomp 8 st_yield 4 st_yresid 2
structural element의 파괴유무 확인명령은 **pl bound stru afail** 임

prop mat 7 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

interface 물성치 입력값으로 블록과 strut가 일체 거동하도록 연결해줌

터널 거동의 정확한 파악을 위하여서는 적당한 값을 주어야하며 이에 대한 물성치 연구가 필요함

본 해석수행의 목적은 단지 허용응력 검토이므로 interface 파괴가 일어나지 않도록 물성치를 크게 주었으며, 주로 해석에 사용되어지는 물성치는 다음과 같으며 생략하는 경우도 있음.

udec 매뉴얼 참고 바람.

if_kn : 1000-10000

if_ks : 1000-10000

if_fric : 45-50

if_ten : 1-100

if_coh : 1-2

property n if_kn if_ks if_friction if_tensile if_cohesion

n 물성번호

if_kn : interface 수직강성

if_ks : interface 전단 변위

if_fric : interface 마찰각

if_ten : interface 인장강도

if_coh : interface 점착력

interface의 파괴 유무 확인 명령어는 **pl bound stru ifail** 임

call rockbolt.fis ; 락볼트 fish 부르기

;상부 중앙

set xCentre = -15.8111 ; x-coord of tunnel centre

set yCentre = 0.5868 ; y-coord of tunnel centre

set theta1 = 40.404704 ; starting angle for cables

set theta2 = 139.595296 ; ending angle for cables

set radius1 = 7.0516 ; radius of tunnel

set radius2 = 11.0516 ; ending radius for remote end of cables(터널반경+락볼트 길이)

set nCables = 9 ; number of cables

rockbolt ; rockbolt fish 실행

;상부 중앙 우측

set xCentre = -14.4037 ; x-coord of tunnel centre

set yCentre = 1.3994 ; y-coord of tunnel centre

set theta1 = 15.499586 ; starting angle for cables

set theta2 = 28.005795 ; ending angle for cables

set radius1 = 5.4264 ; radius of tunnel

set radius2 = 9.4264 ; ending radius for remote end of cables

set nCables = 2 ; number of cables

rockbolt ; rockbolt fish 실행

set xCentre = -16.9335 ; x-coord of tunnel centre

set yCentre = 1.2348 ; y-coord of tunnel centre

set theta1 = 152.452690 ; starting angle for cables

set theta2 = 167.702329 ; ending angle for cables

set radius1 = 5.7556 ; radius of tunnel

set radius2 = 9.75563 ; ending radius for remote end of cables

set nCables = 2 ; number of cables

rockbolt ; rockbolt fish 실행

udec에서 제공하는 fish를 이용하여 록볼트를 모델링하였으며, 케이블 설치 방법은 다음과 같다.

```
cable xc yc xe ye npoint mats asteel matg connect
```

```
xc yc : cable 시작점
```

```
xe ye : cable 끝점
```

```
npoint : segment 수로 4개 이상 줌
```

```
mats : cable steel 물성번호
```

```
asteel : cable 단면적
```

```
matg : grouting 물성번호
```

```
connect : cable을 strut에 최대한 근접시켜줌으로서 일체 거동하도록 해줌  
생각가능 해석에 차이 없음  
단 cable의 segment수를 4개 이상으로 할 경우 임?.
```

udec에서 제공하는 록볼트 설치 fish는 다음과 같다.

```
def rockbolt
```

```
; This example places cable elements along a given arc of tunnel.
```

```
; calculate angle increment between successive cables
```

```
theta1 = degrad * theta1
```

```
theta2 = degrad * theta2
```

```
_angInc = (theta2 - theta1) / float(nCables - 1)
```

```
_ang = theta1
```

```
; get endpoint coordinates
```

```
loop ii (1, nCables)
```

```
_x1 = radius1 * cos(_ang) + xCentre
```

```
_y1 = radius1 * sin(_ang) + yCentre
```

```
_x2 = radius2 * cos(_ang) + xCentre
```

```
_y2 = radius2 * sin(_ang) + yCentre
```

```
; place the cable
```

```
command
```

```
cable _x1 _y1 _x2 _y2 15 8 0.000491 8 ;connect
```

```
;cable xc yc xe ye npoint mats asteel matg connect
```

```
endcommand
```

```
_ang = _ang + _angInc
```

```
endloop
```

```
end
```

```
prop mat 8 cb_dens 7.85e-3 cb_ymod 200e3 cb_yield 250 cb_ycomp 250
```

```
prop mat 8 cb_kb 1.5e3 cb_sb 8
```

해석수행 후 허용응력검토를 위하여 단위중량 및 탄성계수 값은 정확히 주어야하며, cable의 허용응력 및 인장항복강도는 크게 주었음 이는 본 해석 수행의 목적이 단지 허용응력 검토에 있기 때문임. 그러나 파괴시의 거동파악을 위해서는 정확한 물성치를 산정하여 주어야함. 매뉴얼 참고 바람.

```
prop mat=n cb_dens cb_ymod cb_yield cb_ycomp
```

```
cb_dens : cable 단위중량   cb_ymod : cable 탄성계수   cb_yield : cable 인장항복강도   cb_ycomp :  
cable 압축항복강도
```

```
grouting 물성치로 적당한 물성치의 연구가 필요함
```

```
prop mat=n cb_kb cb_sb
```

```
cb_kb : grouting 수직강성
```

```
cb_sb : grouting 전단강성
```

relax ; relax fish 실행

```
;하중분담율 75%적용
def relax
  n=1
  ib=block_head ; start of block list
  loop while ib #0 ; loop through all blocks
    igp=b_gp(ib) ; start of gridpoint list for block ib
    loop while igp # 0 ; loop through all gridpoints
      ibou=gp_bou(igp) ; index of boundary corner associated with gridpoint
      if(ibou) < 0 then ; if address is negative then it is interior
        ibou2=abs(ibou)
        if (imem(ibou2+2)) = 1 then
          imem(ibou2+2)= 1 ; force boundary
          imem(ibou2+3)= 1 ; force boundary
          forcex=ini_force(1,n)
          forcey=ini_force(2,n)
          fmem(ibou2+4)= forcex*0.25 ; apply 25% reaction force
          fmem(ibou2+5)= forcey*0.25 ; apply 25% reaction force
          xforce2=0.0
          yforce2=0.0
          xforce2=fmem(ibou2+4)
          yforce2=fmem(ibou2+5)
          ii=out('ibou2      '+string(ibou2))
          ii=out('n          '+string(n))
          ii=out('xforce0 '+string(forcex))
          ii=out('yforce0 '+string(forcey))
          ii=out('ini_fx   '+string(ini_fx(1,n)))
          ii=out('ini_fy   '+string(ini_fy(1,n)))
          ii=out('xforce2 '+string(xforce2))
          ii=out('yforce2 '+string(yforce2))
          xtable(xforce2,n)=xforce2
          ytable(yforce2,n)=yforce2
          n=n+1
        endif
      endif
      igp=gp_next(igp) ; next gridpoint
    endloop
    ib= b_next(ib) ; next block in list
  endloop
end
```

CYCLE 3000

Save 2.SAV

; <<step 3>>

prop mat 7 st_dens 2.4e-3 st_prat 0.15 st_ymod 1.5e4

prop mat 7 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6

prop mat 7 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

soft 스프링의 물성치를 hard 스프링 물성치로 바꿈

remove :remove fish 실행

;하중분담율 100%적용

```
def remove
  ib=block_head ; start of block list
  loop while ib #0 ; loop through all blocks
    igp=b_gp(ib) ; start of gridpoint list for block ib
    loop while igp # 0 ; loop through all gridpoints
      ibou=gp_bou(igp) ; index of boundary corner associated with gridpoint
      if(ibou) < 0 then ; if address is negative then it is interior
        ibou2=abs(ibou)
        if (imem(ibou2+2)) = 1 then
          imem(ibou2+2)= 1 ; force boundary
          imem(ibou2+3)= 1 ; force boundary
          ini_force(1,n)=0.0
          ini_force(2,n)=0.0
          fmem(ibou2+4)= 0.0 ; apply 0% reaction force
          fmem(ibou2+5)= 0.0 ; apply 0% reaction force
          xforce3=0.0
          yforce3=0.0
          xforce3=fmem(ibou2+4)
          yforce3=fmem(ibou2+5)
          ii=out('ibou2 '+string(ibou2))
          ii=out('xforce3 '+string(xforce3))
          ii=out('yforce3 '+string(yforce3))
          xtable(xforce3,ibou2)=fmem(ibou2+4)
          xtable(yforce3,ibou2)=fmem(ibou2+5)
        endif
      endif
      igp=gp_next(igp) ; next gridpoint
    endloop
    ib= b_next(ib) ; next block in list
  endloop
end
```

CYCLE 3000

Save 3.SAV

* 좌측터널 하부 반단면 굴착

; <<step 4>>

DELETE REGION -22.3 -0.8 -22.7 1.39 -8.93 1.39 -9.37 -0.6

BOUND INTERIOR XVEL=0 YVEL=0 RANGE DOMAIN 9021

CYCLE 1

find

CYCLE 3000

SAVE 4.SAV

; <<step 5>>

STRUCT GEN XC -14.404 YC 1.399 NPOINT 30 MAT 9 THI 0.12 FANG -21.98 THETA 22.98

STRUCT GEN XC -16.934 YC 1.235 NPOINT 30 MAT 9 THI 0.12 FANG 177.36 THETA 23.51

connect명령은 쏫크리트(structure)를 연결시켜주는 명령으로 기존 structure의 끝점과 새로 생기는 structure의 시작점을 연결시켜줌 따라서 우측부와의 연결은 이루어지나 좌측부는 연결이 안됨 따라서 structure를 약간 겹치는 방식으로 하였음. 확인명령은 pl bl stru thick red 임

prop mat 9 st_dens 2.4e-3 st_prat 0.15 st_ymod 0.5e4

prop mat 9 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6

prop mat 9 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

CABLE -22.364 0.862 -26.284 -0.063 15 8 0.000491 8

CABLE -9.202 0.858 -5.302 -0.031 15 8 0.000491 8

prop mat 8 cb_dens 7.85e-3 cb_ymod 200e3 cb_yield 250 cb_ycomp 250

prop mat 8 cb_kb 1.5e3 cb_sb 8

relax

CYCLE 3000

Save 5.SAV

; <<step 6>>

prop mat 9 st_dens 2.4e-3 st_prat 0.15 st_ymod 1.5e4

prop mat 9 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6

prop mat 9 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

remove

CYCLE 3000

Save 6.SAV

* 우측터널 상부 반단면

; <<step 7>>

DELETE REGION 9.089 1.886 10.402 5.640 15.764 7.623 15.764 1.886

DELETE REGION 15.764 1.886 15.764 7.623 21.103 5.596 22.457 1.886

BOUND INTERIOR XVEL=0 YVEL=0 RANGE DOMAIN 22843

CYCLE 1

find

CYCLE 3000

SAVE 7.SAV

; <<step 8>>

STRUCT GEN XC 15.8111 YC 1.4994 NPOINT 100 MAT 10 THI 0.12 FANG 0.01 THETA 179.89

prop mat 10 st_dens 2.4e-3 st_prat 0.15 st_ymod 0.5e4

prop mat 10 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6

prop mat 10 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

;상부 중앙

set xCentre = 15.8111 ; x-coord of tunnel centre

set yCentre = 0.5864 ; y-coord of tunnel centre

set theta1 = 40.404704 ; starting angle for cables

set theta2 = 139.595296 ; ending angle for cables
; (don't wrap back on theta1)

set radius1 = 7.0516 ; radius of tunnel

set radius2 = 11.0516 ; ending radius for remote end of cables

set nCables = 9 ; number of cables

rockbolt

;상부 중앙 우측

set xCentre = 16.9335 ; x-coord of tunnel centre

set yCentre = 1.2348 ; y-coord of tunnel centre

set theta1 = 12.297261 ; starting angle for cables

set theta2 = 27.547310 ; ending angle for cables
; (don't wrap back on theta1)

set radius1 = 5.7556 ; radius of tunnel

set radius2 = 9.7556 ; ending radius for remote end of cables

set nCables = 2 ; number of cables

rockbolt

```

;상부 중앙 좌측
set xCentre = 14.4037      ; x-coord of tunnel centre
set yCentre = 1.3994      ; y-coord of tunnel centre
set theta1 = 152.604868   ; starting angle for cables
set theta2 = 168.801134   ; ending angle for cables
                             ; (don't wrap back on theta1)
set radius1 = 5.4264      ; radius of tunnel
set radius2 = 9.4264      ; ending radius for remote end of cables
set nCables = 2           ; number of cables
rockbolt

prop mat 8 cb_dens 7.85e-3 cb_ymod 200e3 cb_yield 250 cb_ycomp 250
prop mat 8 cb_kb 1.5e3 cb_sb 8

relax

CYCLE 3000
Save 8.SAV

;<<step 9>>
prop mat 10 st_dens 2.4e-3 st_prat 0.15 st_ymod 1.5e4
prop mat 10 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6
prop mat 10 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

remove

CYCLE 3000
Save 9.sav

*****
* 우측터널 하부 반단면 굴착
*****

;<<step 10>>
DELETE REGION 13.819 -0.6 9.089 1.886 22.457 1.886 21.681 -0.782

BOUND INTERIOR XVEL=0 YVEL=0 RANGE DOMAIN 1083674
CYCLE 1

find

CYCLE 3000
SAVE 10.SAV

```

; <<step 11>>

STRUCT GEN XC 14.404 YC 1.399 NPOINT 30 MAT 12 THI 0.12 FANG 178 THETA 22.98
STRUCT GEN XC 16.934 YC 1.235 NPOINT 30 MAT 12 THI 0.12 FANG -21.01 THETA 23.51

prop mat 12 st_dens 2.4e-3 st_prat 0.15 st_ymod 0.5e4
prop mat 12 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6
prop mat 12 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

CABLE 22.364 0.862 26.284 0.063 15 8 0.000491 8
CABLE 9.202 0.858 5.302 -0.301 15 8 0.000491 8

prop mat 8 cb_dens 7.85e-3 cb_ymod 200e3 cb_yield 250 cb_ycomp 250
prop mat 8 cb_kb 1.5e3 cb_sb 8

relax

CYCLE 3000
Save 11.SAV

; <<step 12>>

prop mat 12 st_dens 2.4e-3 st_prat 0.15 st_ymod 1.5e4
prop mat 12 st_ycomp 20.6 st_yield 20.6 st_yresid 20.6
prop mat 12 if_kn 1.0e5 if_ks 1.0e5 if_fric= 50 if_ten 2.06 if_coh 100

remove

CYCLE 3000
save 12.SAV

set log off

[결과정리]

그림파일 출력

```
res 파일명.sav
set back 7
set pl dxf ; 또는 set pl bmp color 명령을 사용하여 bmp그림파일로 출력가능
set out mesh.dxf ; set out mesh.bmp
pl pen bl red

new
res 파일명.sav
win -35 35 -25 45 ; x1 x2 y1 y2 로 정사각형 영역으로 잡아야 그림 옆에 여백이 생김
```

*****터널 주변지반의 응력 및 변위도*****

;블락의 변위도

```
set out dis.dxf
pl pen bl red dis blue
```

```
set out xdis.dxf
pl pen bl red xdis blue
```

```
set out ydis.dxf
pl pen bl red ydis blue
```

;지반응력

```
set out stru.dxf
pl pen bl red stress blue
```

;plastic 영역

```
set out plas.dxf
pl pen bl red pl
```

;shear displacement

```
set out sheardis.dxf
pl pen bl red sh blue
```

*****S/C의 결과 출력*****

```
set out scax.dxf
pl pen bl red stru axial blue
```

```
set out scsh.dxf
pl pen bl red stru she blue
```

```
set out scmo.dxf
pl pen bl red stru mom blue
```

*****R/B의 결과 출력*****

```
set out rockbolt.dxf
pl pen bl red ca cable axial blue
```


변위 log파일 출력

```
set log dis.log  
set log on
```

```
res 파일명.sav  
pri his 2  
pri hmax 2
```

```
pri his 3  
pri hmax 3
```

```
pri his 4  
pri hmax 4
```

```
pri his 5  
pri hmax 5
```

```
.....  
.....  
.....  
.....  
set log off
```

췁크리트응력 log파일 출력

```
set log stru.log  
set log on  
res 파일명.sav
```

```
pr stru  
set log off
```

;결과물을 엑셀에서 정리 및 udec에서 제공하는 fish문을 이용하여 정리

록볼트응력 log파일 출력

```
set log cable.log  
set log on  
res 파일명.sav
```

```
pr cable  
set log off
```

;udec 화면상에서 확인하는 것이 편함

변위출력시 아래의 fish문을 이용하면 편리함

; 좌표 집어 넣구서 이 파일을 부르면 자동으로 disp.log라는 파일이 생김

그 파일을 보면 천단변위 하구 내공변위 볼 수 있음

new

res 파일명.sav

def tun_dis

;좌측터널

upper_xpos = -15.8111 ; 여기에다가 천단변위 측정할 좌표 넣음됨

upper_ypos = 7.6484

left_xpos = -22.6967 ; 여기에다가 내공변위 측정할 좌표 넣음됨

left_ypos = 1.4994

right_xpos = -8.9673 ; 여기에다가 내공변위 측정할 좌표 넣음됨

right_ypos = 1.4994

;우측터널

a_xpos =15.8111 ; 여기에다가 천단변위 측정할 좌표 넣음됨

a_ypos =7.6484

b_xpos =8.9673 ; 여기에다가 내공변위 측정할 좌표 넣음됨

b_ypos =1.4994

c_xpos = 22.6967 ; 여기에다가 내공변위 측정할 좌표 넣음됨

c_ypos = 1.4994

pdisp1=gp_near(upper_xpos, upper_ypos)

pdisp2=gp_near(left_xpos, left_ypos)

pdisp3=gp_near(right_xpos, right_ypos)

pdisp4=gp_near(a_xpos, a_ypos)

pdisp5=gp_near(b_xpos, b_ypos)

pdisp6=gp_near(c_xpos, c_ypos)

m_crown = (gp_ydis(pdisp1))*1000

m_left = (gp_xdis(pdisp2))*1000

m_right = (gp_xdis(pdisp3))*1000

c_crown = (gp_ydis(pdisp4))*1000

c_left = (gp_xdis(pdisp5))*1000

c_right = (gp_xdis(pdisp6))*1000

command

set log disp.log

set log on

end_command

ii=out('m_crown = '+string(m_crown))

ii=out('m_left = '+string(m_left))

ii=out('m_right = '+string(m_right))

ii=out('c_crown = '+string(c_crown))

ii=out('c_left = '+string(c_left))

ii=out('c_right = '+string(c_right))

command

set log off

end_command

end

tun_dis ; fish문 실행명령

췁크리트 응력계산 fish로 struc.dat 문을 다음과 같이 만든다.

```
res 파일명.sav
call prstruc.fis
set echo on
set b_space=1.0 b_height=0.12
set log pr_struc.log
set log on
pr_struc ; pr_struc의 함수명 실행
set log off
```

prstruc.fis 문은 다음과 같다.

```
ca str.fin ; udec에서 제공하는 fin문으로 계산 폴더에 넣어져야함
def pr_struc
;-----
; fax : actual axial force
; m1 : moment end 1
; m2 : moment end 2
; stax : component of axial stress due to axial force
; stm : component of axial stress due to moment at beam mid-point
; stmin: minimum stress at beam mid-point
; stmax: maximum stress at beam mid-point
; Note : stresses are positive in tension and stmin < stmax
;-----
ipnt = str_elem_head
ii = out(' ID          F-axial          Str-min          Str-max  ')
ii = out('----  -----  -----  -----  ')
loop while ipnt # 0
  id   = imem(ipnt+$SELID)
  do2i = b_height * 0.5 / fmem(ipnt+$SELIN);단면계수 Z
  bacs = fmem(ipnt+$SELARS)
  fax  = fmem(ipnt+$SELFAX) * b_space
  m1   = fmem(ipnt+$SELM1) * b_space
  m2   = fmem(ipnt+$SELM2) * b_space
  stax = - fax / bacs
  stm  = (abs(m2 - m1)*0.5*do2i) ; m1과 m2의 평균값을 이용 모멘트에 의한 응력계산
  stmin = (stax - stm)*10.1972 ; kg/cm^2단위로 환산
  stmax = (stax + stm)*10.1972 ; kg/cm^2단위로 환산
  st1   = string(id)
  st2   = string(fax)
  st3   = string(m1)
  st4   = string(m2)
  st5   = string(stmin)
  st6   = string(stmax)
  ii   = out(' '+st1+' '+st2+' '+st5+' '+st6)
  ipnt = imem(ipnt)
end_loop
end
```

자주 쓰는 결과 출력 명령어

```
pl dn
pl cable axial
pl stru axial
pl sig 1
pl sig 2
pl dis
pl xdis
pl ydis
pl bl slip lmag open yell
plot bl bound xcond
plot bl bound ycond
pl bl vel
pl bl xvel
pl bl yvel
pl zones
pl bl pla
print max
pl bl overlap
pl bl contacts
pl bl mohr
pl bl gp
pl stru ifail
pl stru afail
```

해석수행도중 결과확인 후 다시 해석을 수행하고자 할 경우 ESC키로 해석수행을 멈추고 결과확인 후
solve f=0.001 식으로 입력하면 됨